Data Visualization: Introduction to gnuplot

Basic functions

- See here for installation instructions and the basic functionalities of gnuplot.

```
Function            Returns
      -----------      ------------------------------------
      abs(x)            absolute value of x, |x|
      acos(x)           arc-cosine  of x
      asin(x)           arc-sine    of x
      atan(x)           arc-tangent of x
      cos(x)            cosine      of x,  x is in radians.
      cosh(x)           hyperbolic cosine of x, x is in radians
      erf(x)            error function of x
      exp(x)            exponential function of x, base e
      inverf(x)         inverse error function of x
      invnorm(x)        inverse normal distribution of x
      log(x)            log of x, base e
      log10(x)          log of x, base 10
      norm(x)           normal Gaussian distribution function
      rand(x)           pseudo-random number generator
      sgn(x)            1 if x > 0, -1 if x < 0, 0 if x=0
      sin(x)            sine      of x, x is in radians
      sinh(x)           hyperbolic sine of x, x is in radians
      sqrt(x)           the square root of x
      tan(x)            tangent of x, x is in radians
      tanh(x)           hyperbolic tangent of x, x is in radians
```

- One can also plot complex functions like **`sin(x)/x`**.

Plotting data files

```
gnuplot> p "Q_t_K7L98M8.0000.dat" using 1:2 w l t'sample' lw 2 lc -1 lt 1
gnuplot> set log x
gnuplot> p "Q_t_K7L98M8.0000.dat" using 1:2 w l t'sample' lw 2 lc -1 lt 1
gnuplot> p "filledK7L98M7.6000_t" u 1:3 w l lw 2
gnuplot> p "filledK7L98M7.6000_t" u 1:4 w l lw 2
gnuplot> !head filledK7L98M7.6000_t
#t rho abs(m) m
0 9.336735e-01 5.102041e-03 -5.102041e-03
50000 9.569971e-01 1.508746e-01 -1.508746e-01
100000 9.482507e-01 2.937318e-01 -2.937318e-01
150000 9.569971e-01 3.272595e-01 -3.272595e-01
200000 9.548105e-01 2.871720e-01 -2.871720e-01
250000 9.613703e-01 2.223032e-01 -2.223032e-01
300000 9.511662e-01 1.115160e-01 -1.115160e-01
350000 9.635569e-01 1.034985e-01 -1.034985e-01
400000 9.555394e-01 1.333819e-01 -1.333819e-01
gnuplot> p "filledK7L98M7.6000_t" u 1:2 w l lw 2
gnuplot> p "filledK7L98M7.6000_t" u 1:4 w l lw 2
gnuplot> p "filledK7L98M7.6000_t" u 3:4 w l lw 2
```

Plotting at field intervals: `p "Q_t_K7L98M8.0000.dat" using 1:2 every 50 pt 10 t'sample'`

Generating eps figure

```
rumelalahiri$ pwd
/Users/rumelalahiri/Desktop/course2020/comp_phys/test

rumelalahiri$ cat dynamics.gnu
set term postscript eps enhanced "Times-Roman" 26 color
set out 'dynamics.eps'
set size 1,1
set size square
set log x
set ytics 0.1
set xrange[:]
#set format x "10^{%T}"
set format x "%3.2t{/Symbol \264}10^{%L}"
#set format x "%2.0t{/Symbol \327}10^{%L}"
#set format x "%2.0t{/Symbol \261}10^{%L}"
set yrange[0.5:1]
set key bottom left
c=1.0
set label "L=98" at 1e5,0.85
set label '{/Times-Roman=28 test}' at 1e5,0.75
#set key at 0.6,8e-2
set key samplen 1.2
set pointsize 1.3
set key font ",18"
set key spacing 1.5
#unset key
#rescaling: $2/c
p "Q_t_K7L98M7.0000_.dat" u 1:($2/c) w l lw 4 lc rgb '#f0027f' t'data01',\
"Q_t_K7L98M8.0000.dat" u 1:($2/c) w lp pt 6 lw 2 lc rgb '#386cb0' t'data02'
```

- The syntax for the digit is "%" + (total length).(precision)
- "%t" represents the mantissa to base 10
- "%L" represents the power to base 10.
- For format A x 10^B, A is given by "%t" and B is "%L"

EPS is a PostScript image file format. EPS files contain PostScript code, which is used for storing font and vector image information

*__PostScript__ is a page description language (PDL) that describes a page's text and graphical content

List of Symbols

PostScript Character Codes

T = text (here Times-Roman) S = Symbol Z = ZapfDingbats E = ISO Latin-1 encoding
(the "E" character set is accessed via an option on "set encoding")

	T	S	Z	E		T	S	Z	E		T	S	Z	E		T	S	Z	E		T	S	Z	E
040					111	I	I	☆	I	162	r	ρ	❒	r	256	fi	→	③	®	327		·	↕	×
041	!	!	✂	!	112	J	ϑ	✪	J	163	s	σ	▲	s	257	fl	↓	④	¯	330		¬	↘	Ø
042	"	∀	✁	"	113	K	K	☆	K	164	t	τ	▼	t	260		°	⑤	°	331		∧	→	Ù
043	#	#	✃	#	114	L	Λ	✡	L	165	u	υ	◆	u	261	–	±	⑥	±	332		∨	↗	Ú
044	$	∃	✄	$	115	M	M	★	M	166	v	ϖ	❖	v	262	†	″	⑦	²	333		⇔	➔	Û
045	%	%	☎	%	116	N	N	✡	N	167	w	ω	◗	w	263	‡	≥	⑧	³	334		⇐	➜	Ü
046	&	&	✆	&	117	O	O	✯	O	170	x	ξ	❙	x	264	·	×	⑨	´	335		⇑	→	Ý
047	’	∍	✇	’	120	P	Π	✰	P	171	y	ψ	❚	y	265		∝	⑩	μ	336		⇒	➞	Þ
050	(	(	✈	(	121	Q	Θ	✱	Q	172	z	ζ	❯	z	266	¶	∂	❶	¶	337		⇓	➟	ß
051	)	)	✉	)	122	R	P	✲	R	173	{	{	❛	{	267	•	•	❷	·	340		◊	➠	à
052	*	*	☛	*	123	S	Σ	✳	S	174					270	,	÷	❸	¸	341	Æ	⟨	➡	á
053	+	+	☞	+	124	T	T	✴	T	175	}	}	❝	}	271	„	≠	❹	¹	342		®	➢	â
054	,	,	✌	,	125	U	Y	✵	U	176	~	~	❞	~	272	”	≡	❺	º	343	ª	©	➢	ã
055	-	–	✍	–	126	V	ς	✶	V	220				ı	273	»	≈	❻	»	344		™	➤	ä
056	.	.	✎	.	127	W	Ω	✷	W	221				`	274	…	…	❼	¼	345		∑	➥	å
057	/	/	✏	/	130	X	Ξ	✸	X	222				´	275	‰	|	❽	½	346		⎛	➦	æ
060	0	0	✐	0	131	Y	Ψ	✹	Y	223				ˆ	276		—	❾	¾	347		⎜	➧	ç
061	1	1	✑	1	132	Z	Z	✺	Z	224				˜	277	¿	↲	❿	¿	350	Ł	\	➨	è
062	2	2	✒	2	133	[	[	✻	[	225				¯	300		ℵ	①	À	351	Ø	⎡	➩	é
063	3	3	✓	3	134	\	∴	✼	\	226				˘	301	`	ℑ	②	Á	352	Œ	⎢	➪	ê

Different line types, point types, and error bars




```
gnuplot> set pointsize 2.0
gnuplot> p "linept.dat" u 1:2 w lp pt 7 lw 3 lc rgb '#d8b365' t'',\
"linept_.dat" u 1:2 w lp pt 4 lw 3 lc rgb '#5ab4ac' t''


gnuplot> p "linept.dat" u 1:2:3 w err pt 7 lw 3 lc rgb '#d8b365',\
"linept.dat" u 1:2 w l lc -1 lw 3
gnuplot> p "linept.dat" u 1:($2-$3):($2+$3) with filledcurves lc rgb '#d8b365',\
"linept.dat" u 1:2 w l lc -1 lw 3
```

ColorBrewer

[This](colorbrewer2.org/#type=sequential&scheme=RdPu&n=7) is a very good webpage to choose colors for your plots!



Sequential color scheme




```
lc rgb ‘#XXXXXX’
```

Creating a plot with inset

```
set term postscript eps enhanced "Times-Roman" 26 color

set out 'fig_inset.eps'
set multiplot
set size 1,1
set ytics 0.5
#set origin 0.1,0.1
plot sin(x) lw 3 t''
set size 0.45,0.45
set origin 0.5,0.5
#set autoscale
set ytics 200
set tics font "Times-Roman, 12"
set xrange[-5:5]
plot (x*x*x*x*x) lw 3 lc 2 t'',x**3.0 lw 3 lc -1 t''
unset multiplot
```



Fitting data with gnuplot

```
gnuplot> f(x)= A+B*x
gnuplot> fit f(x) 'fit_data.txt' u 1:2 via A,B
```

```
Final set of parameters            Asymptotic Standard Error
=======================            ==========================
A                  = 0.848014      +/- 0.01284      (1.514%)
B                  = -0.0388181    +/- 0.002229     (5.742%)
```



Simple 3d plot

```
set key off
splot 'xyz.tsv' using 1:2:3 w points palette pointsize 3 pt 7
```



Palette Mapped surface View

```
set view map
set size ratio .9
splot "xyz.tsv" using 1:2:3 w points pt 5 pointsize 1 palette lw 30
```

3d surface plot

- **pm3d** is a style for drawing palette-mapped 3d data as color maps and surfaces. It uses a pm3d algorithm which allows plotting gridded as well as non-gridded data without preprocessing.

```
set terminal pngcairo font "arial, 10" fontscale 1.0 size 600, 400
set output 'pm3d.1.png'
set pm3d
set samples 40, 40
set isosamples 20, 20
set title "pm3d demo. Radial sinc function. Default options."
set xlabel "x"
set xrange [-15.0000: 15.0000]
set x2range [*: *]
set ylabel "y"
set yrange [-15.0000: 15.0000]
set y2range [*: *]
set zrange [-0.250000: 1.00000]
set cbrange[*:*]
set palette defined (-1 "red", 0 "white", 1 "blue")
splot sin (sqrt (x ** 2 + y ** 2)) / sqrt (x ** 2 + y ** 2)
```

- **set cbrange** command sets the range of values which are colored using the current palette
- **samples** is used to set the number of function evaluations along an axis in the range being plotted.
- The **isoline** density for plotting functions. **isoline** is a curve parameterized by one of the surface parameters while the other parameter is fixed. Isolines provide a simple means to display a surface.

3d surface plot

- This is the output of the script presented in the earlier slide.



Color plot



unset surface in splot will not draw points or lines corresponding to the function or data file points

```
set term postscript eps enhanced "Times-Roman" 26 color
set out 'color_plot.eps'
set size 1,1
unset key
set size ratio 0.8
set view map
set xtics 0.01
set ytics 0.1
set pm3d
set palette rgb 23,28,3
unset surface
splot [0.005:]"color_plot.dat" u 2:1:3
```